

**mestreg postestimation** — Postestimation tools for mestreg

[Postestimation commands](#)
[predict](#)
[margins](#)
[Remarks and examples](#)
[Methods and formulas](#)
[Also see](#)

## Postestimation commands

The following postestimation commands are of special interest after `mestreg`:

Command	Description
<code>stcurve</code>	plot the survivor, hazard, and cumulative hazard functions
<code>estat group</code>	summarize the composition of the nested groups

The following standard postestimation commands are also available:

Command	Description
<code>contrast</code>	contrasts and ANOVA-style joint tests of estimates
<code>estat ic</code>	Akaike's and Schwarz's Bayesian information criteria (AIC and BIC)
<code>estat summarize</code>	summary statistics for the estimation sample
<code>estat vce</code>	variance–covariance matrix of the estimators (VCE)
<code>estat (svy)</code>	postestimation statistics for survey data
<code>estimates</code>	cataloging estimation results
* <code>hausman</code>	Hausman's specification test
<code>lincom</code>	point estimates, standard errors, testing, and inference for linear combinations of coefficients
* <code>lrtest</code>	likelihood-ratio test
<code>margins</code>	marginal means, predictive margins, marginal effects, and average marginal effects
<code>marginsplot</code>	graph the results from margins (profile plots, interaction plots, etc.)
<code>nlcom</code>	point estimates, standard errors, testing, and inference for nonlinear combinations of coefficients
<code>predict</code>	predictions, residuals, influence statistics, and other diagnostic measures
<code>predictnl</code>	point estimates, standard errors, testing, and inference for generalized predictions
<code>pwcompare</code>	pairwise comparisons of estimates
<code>test</code>	Wald tests of simple and composite linear hypotheses
<code>testnl</code>	Wald tests of nonlinear hypotheses

\* `hausman` and `lrtest` are not appropriate with `svy` estimation results.

# predict

## Description for predict

`predict` creates a new variable containing predictions such as mean and median survival times, hazards, survivor functions, linear predictions, and standard errors.

## Menu for predict

Statistics > Postestimation

## Syntax for predict

*Syntax for obtaining predictions of the outcome and other statistics*

```
predict [type] newvarsspec [if] [in] [, statistic options]
```

*Syntax for obtaining estimated random effects and their standard errors*

```
predict [type] newvarsspec [if] [in], reffects [re_options]
```

*Syntax for obtaining ML scores*

```
predict [type] newvarsspec [if] [in], scores
```

*newvarsspec* is *stub\** or *newvarlist*.

<i>statistic</i>	Description
Main	
<code>mean</code>	mean survival time; the default
<code>median</code>	median survival time
<code>hazard</code>	hazard
<code>eta</code>	fitted linear predictor
<code>xb</code>	linear predictor for the fixed portion of the model only
<code>stdp</code>	standard error of the fixed-portion linear prediction
<code>surv</code>	predicted survivor function
<code>density</code>	predicted density function
<code>distribution</code>	predicted distribution function

These statistics are available both in and out of sample; type `predict ... if e(sample) ...` if wanted only for the estimation sample.

<i>options</i>	Description
Main	
<u>conditional</u> ( <i>ctype</i> )	compute <i>statistic</i> conditional on estimated random effects; default is <code>conditional(ebmeans)</code>
<u>marginal</u>	compute <i>statistic</i> marginally with respect to the random effects
<u>nooffset</u>	make calculation ignoring offset or exposure
Integration	
<u>int_options</u>	integration options
median may not be combined with marginal.	
<i>ctype</i>	Description
<u>ebmeans</u>	empirical Bayes means of random effects; the default
<u>ebmodes</u>	empirical Bayes modes of random effects
<u>fixedonly</u>	prediction for the fixed portion of the model only
<i>re_options</i>	Description
Main	
<u>ebmeans</u>	use empirical Bayes means of random effects; the default
<u>ebmodes</u>	use empirical Bayes modes of random effects
<u>reses</u> ( <i>stub*</i>   <i>newvarlist</i> )	calculate standard errors of empirical Bayes estimates
Integration	
<u>int_options</u>	integration options
<i>int_options</i>	Description
<u>intpoints</u> (#)	use # quadrature points to compute marginal predictions and empirical Bayes means
<u>iterate</u> (#)	set maximum number of iterations in computing statistics involving empirical Bayes estimators
<u>tolerance</u> (#)	set convergence tolerance for computing statistics involving empirical Bayes estimators

## Options for predict

Main

`mean`, the default, calculates the mean survival time.

`median` calculates the median survival time.

`hazard` calculates the hazard. When `marginal` is specified, marginal hazard is calculated as a ratio of the marginal density to the marginal survivor function.

`surv` calculates the predicted survivor function.

`eta`, `xb`, `stdp`, `density`, `distribution`, `scores`, `conditional()`, `marginal`, and `nooffset`; see [ME] [mestreg postestimation](#). `marginal` may not be specified with `median`.

reffects, ebmeans, ebmodes, and reses()); see [ME] [meglm postestimation](#).

Integration

intpoints(), iterate(), and tolerance()); see [ME] [meglm postestimation](#).

## margins

### Description for margins

margins estimates margins of response for mean and median survival times and linear predictions.

### Menu for margins

Statistics > Postestimation

### Syntax for margins

```
margins [marginlist] [, options]
```

```
margins [marginlist] , predict(statistic ...) [predict(statistic ...) ...] [options]
```

<i>statistic</i>	Description
mean	mean survival time; the default
median	median survival time
xb	linear predictor for the fixed portion of the model only
hazard	not allowed with margins
eta	not allowed with margins
stdp	not allowed with margins
surv	not allowed with margins
<u>density</u>	not allowed with margins
<u>distribution</u>	not allowed with margins
reffects	not allowed with margins
scores	not allowed with margins

Options conditional(ebmeans) and conditional(ebmodes) are not allowed with margins.

Option marginal is assumed where applicable if conditional(fixedonly) is not specified.

Statistics not allowed with margins are functions of stochastic quantities other than  $e(b)$ .

For the full syntax, see [R] [margins](#).

## Remarks and examples

[stata.com](http://www.stata.com)

Various predictions, statistics, and diagnostic measures are available after fitting a mixed-effects parametric survival model with `mestreg`. For the most part, predictions center on obtaining estimates of the survival times or hazard functions. Conditional predictions are based on the computation of the group-specific random effects, and marginal predictions are obtained by numerically integrating out the random effects.

## ▷ Example 1

In [example 1](#) of [ME] `mestreg`, we analyzed the time to infection of the catheter insertion point for 38 kidney dialysis patients. We fit the following model:

```
. use http://www.stata-press.com/data/r14/catheter
(Kidney data, McGilchrist and Aisbett, Biometrics, 1991)
. stset time, failure(infect)
(output omitted)
. mestreg age female || patient:, distribution(weibull)
(output omitted)
```

The `predict` command allows us to compute marginal and conditional predictions. Unless stated differently, we use the word “conditional” to mean “conditional on the empirical Bayes predictions of the random effects”. Below we compute marginal and conditional means for the mean survival time.

```
. predict m_marg, mean marginal
. predict m_cond, mean conditional
(predictions based on fixed effects and posterior means of random effects)
(using 7 quadrature points)
```

Now, we can display the predictions for some of the patients.

```
. sort female age patient
. list patient female age m_* in 15/20, sepby(patient)
```

	patient	female	age	m_marg	m_cond
15.	29	0	53	52.7936	22.36021
16.	29	0	53	52.7936	22.36021
17.	16	0	60	50.67549	28.01287
18.	16	0	60	50.67549	28.01287
19.	38	0	60	50.67549	49.47013
20.	38	0	60	50.67549	49.47013

We see in the output that the predicted expected conditional mean for patient 29 is equal to 22.36 (shown in `m_cond`). This is the expected time to infection for this patient. However, the predicted marginal mean for this patient is 52.79 (shown in `m_marg`). This is the expected time to infection for a patient from the population who is male and is 53 years old. This particular patient seems to be more prone to infection than would be expected based on his age and gender.


Conditional predictions are specific to each group, while marginal predictions are the same within each covariate pattern through the data. Patients 16 and 38 have the same covariate patterns; therefore, their marginal predicted means are the same. However, conditional predicted means differ.

margins and marginsplot show the changes in the marginal means for different ages.

```
. margins, predict(mean marginal) at(female=0 age=(20(5)70)) noatlegend
Adjusted predictions          Number of obs      =          76
Model VCE      : OIM
Expression    : Marginal predicted mean, predict(mean marginal)
```

	Delta-method					[95% Conf. Interval]	
	Margin	Std. Err.	z	P> z			
_at							
1	64.03488	28.99891	2.21	0.027	7.19805	120.8717	
2	62.1891	26.33293	2.36	0.018	10.5775	113.8007	
3	60.39652	24.11465	2.50	0.012	13.13268	107.6604	
4	58.65562	22.37009	2.62	0.009	14.81105	102.5002	
5	56.96489	21.11496	2.70	0.007	15.58032	98.34946	
6	55.3229	20.34546	2.72	0.007	15.44653	95.19927	
7	53.72824	20.03199	2.68	0.007	14.46626	92.99023	
8	52.17955	20.12007	2.59	0.010	12.74494	91.61415	
9	50.67549	20.53858	2.47	0.014	10.42062	90.93037	
10	49.21479	21.2114	2.32	0.020	7.641204	90.78838	
11	47.7962	22.06721	2.17	0.030	4.545267	91.04712	

```
. marginsplot
Variables that uniquely identify margins: age
```



We see that the predicted marginal mean decreases with age; older patients are expected to have an event earlier. This is consistent with the findings from [example 1](#) of [ME] `mestreg` that the hazard is increasing with age.

## Example 2

Continuing with [example 1](#), we now predict survivor functions.


```
. predict S_marg, surv marginal
(using 7 quadrature points)
. predict S_cond, surv conditional
(predictions based on fixed effects and posterior means of random effects)
(using 7 quadrature points)
. sort female age patient _t
. list patient female age _t S_* in 15/20, sepy(patient)
```

	patient	female	age	_t	S_marg	S_cond
15.	29	0	53	2	.9628581	.9564018
16.	29	0	53	25	.516502	.3493612
17.	16	0	60	4	.9122224	.9230725
18.	16	0	60	17	.62736	.6129258
19.	38	0	60	8	.8141541	.9107043
20.	38	0	60	63	.2048696	.2900459

Survival predictions vary with the value of the study time variable because they are predictions of the survivor function at the study time `_t`. For example, patient 29 has a 0.96 probability that a new insertion remains at least 2 days without infection and a 0.35 probability that a new insertion remains at least 25 days without infection. For a random patient from the population, the probabilities to remain at least 2 or 25 days without infection are, respectively, 0.96 and 0.52.

We can use `stcurve` to plot these predictions simultaneously for males and females of the same age.

```
. stcurve, surv at1(female=0 age=53) at2(female=1 age=53)
(option unconditional assumed)
```




We see that the survivor function for females is above the survivor function for males, which means that females have a greater probability of not having an episode by study time `_t`.

### ► Example 3

In [example 2](#) of [ME] `mestreg`, we estimated two different distributions with random effects on patient and covariates age and female. Here we compare the marginal hazards using `stcurve`. By default, `stcurve` plots predictions at the mean of the covariates, computed over the whole estimation sample. We plot the predictions for `female==1`.

```
. mestreg age female || patient:, dist(weibull) time
  (output omitted)
. stcurve, hazard at(female=1)
(option unconditional assumed)
. graph save g1
(file g1.gph saved)
. mestreg age female || patient:, dist(gamma)
  (output omitted)
. stcurve, hazard at(female=1)
(option unconditional assumed)
. graph save g2
(file g2.gph saved)
. graph combine g1.gph g2.gph
```



The two estimated marginal hazards are similar. The marginal hazard has a very different shape from the conditional hazards. The conditional hazard function for a Weibull or a gamma distribution are both monotonic (increasing, constant, or decreasing, depending on the parameters).




## ▷ Example 4

In [example 3](#) of [ME] `mestreg`, we fit a Weibull model with random intercepts and random coefficients at the subject level. We obtained a positive covariance between the random effects. We refit the model here and then use `predict` with the option `reffects` to obtain predictions of the random effects based on the empirical Bayes posterior means.

```
. use http://www.stata-press.com/data/r14/angina, clear
(Angina drug data, Rabe-Hesketh and Skrondal, 2012, ch 15.7)
. mestreg occasion##treat || pid: treat, distribution(weibull)
> covariance(unstructured)
(output omitted)
. predict re*, reffects
(calculating posterior means of random effects)
(using 7 quadrature points)
```

Plotting the predictions of the predicted random coefficient versus the random intercept shows the pattern we discussed in the main section: individuals with a larger random slope tend also to have a larger random intercept.

```
. twoway scatter re1 re2
```



Individuals with large random intercepts have individual hazards that are larger than those of other individuals with the same covariate patterns. Also, individuals with large random coefficients have individual conditional hazard ratios for treatment that are larger than those of other individuals with the same covariate pattern.

In other words, if the aim of the treatment is to decrease the hazard, then the positive correlation means that the treatment tends to be less effective for individuals who have a higher individual hazard (within the same occasion number).

◀

## ▷ Example 5

In [example 1](#) of [ME] `mestreg`, we mentioned that hazard ratios should be interpreted as conditional on the random effects. Here we use `predict` to illustrate this concept. We use a simulated dataset for a Weibull model with random effects for `group` and a binary covariate `x`.

We show that for a given group, the conditional hazard function satisfies the proportional-hazards (PH) assumption. That is, for a given group  $j$ ,

$$h(t|x = 1, \text{group} = j) = \exp(\beta_x) \times h(t|x = 0, \text{group} = j)$$

is equivalent to

$$\log\{h(t|x = 1, \text{group} = j)\} = \beta_x + \log\{h(t|x = 0, \text{group} = j)\}$$

This property of the log hazard-function translates to one curve being a shifted version of the other, which is easier to see than the proportionality of the (untransformed) hazard function.

After fitting the model, we use `predict` to compute the conditional prediction of the hazard function for group 1; we create the variables `hcond0` and `hcond1`. `hcond0` will contain the conditional hazard for group 1 when `x==0`; `hcond1` will contain the conditional hazard for group 1 when `x==1`.

We also create `zcond = loghcond0 +  $\beta_x$` . If the PH assumption is satisfied, then the plotted values of `zcond` will be superimposed on those of `loghcond1`.


```
. use http://www.stata-press.com/data/r14/weibre, clear
. mestreg i.x || group:, distribution(weibull) nolog
      failure _d: 1 (meaning all fail)
      analysis time _t: t
Mixed-effects Weibull regression          Number of obs    =    100,000
Group variable:                group      Number of groups =         500
                                      Obs per group:
                                      min =         200
                                      avg =        200.0
                                      max =         200
Integration method: mvaghermite          Integration pts.  =         7
                                      Wald chi2(1)      =    21447.86
Log likelihood = -228103.73              Prob > chi2      =         0.0000
```

_t	Haz. Ratio	Std. Err.	z	P> z	[95% Conf. Interval]	
1.x	2.713137	.0184908	146.45	0.000	2.677137	2.749622
_cons	2.564135	.0797383	30.28	0.000	2.412518	2.72528
/ln_p	-.6925791	.0024746	-279.88	0.000	-.6974291	-.687729
group var(_cons)	.4728021	.0303093			.4169774	.5361006

```
LR test vs. Weibull model: chibar2(01) = 35800.39      Prob >= chibar2 = 0.0000
```

```
. predict hcond, hazard conditional(ebmeans)
(predictions based on fixed effects and posterior means of random effects)
. gen loghcond0 = log(hcond) if x==0
(49,991 missing values generated)
. gen loghcond1 = log(hcond) if x==1
(50,009 missing values generated)
. gen zcond = loghcond0 + _b[_t:1.x]
(49,991 missing values generated)
. sort _t group
```


```
. twoway line loghcond0 loghcond1 zcond _t if group==1
```



In the graph above, the line for `loghcond1` cannot be distinguished from the line for `zcond` for most of the distribution. This illustrates that the PH assumption is satisfied for the conditional hazard. Notice that you can still see a part of `loghcond1` near the origin. This is because the two variables correspond to different values of `_t` and only `loghcond1` happens to be defined at the early values.

Now, we make the same computation for the marginal hazard.

```
. predict hmarg, hazard marginal
. gen loghmarg0 = log(hmarg) if x==0
(49,991 missing values generated)
. gen loghmarg1 = log(hmarg) if x==1
(50,009 missing values generated)
. gen zmarg = loghmarg0 + _b[_t:1.x]
(49,991 missing values generated)
. sort _t group
. twoway line loghmarg0 loghmarg1 zmarg _t
```



The curve for `zmargin` is clearly different from the curve for `loghmargin1`, demonstrating that the marginal distribution does not meet the PH assumption. Notice that the line for `loghmargin1` is shorter than the others. This is because predictions are obtained at the values of `_t` in the dataset. These values of `_t` were simulated based on the model, which determines that observations with `x==1` fail earlier.

◀

## Methods and formulas

Methods and formulas for predicting random effects and other statistics are given in [Methods and formulas](#) of [ME] **meglm postestimation**. Statistics of special interest for survival analysis are described below.

`predict newvar` with the `conditional()` option computes the following predictions:

median:

$$newvar_{ji} = \{t : \widehat{S}(t|\mathbf{x}_{ji}, \widehat{u}_{ji}) = 1/2\}$$

where  $\widehat{S}(t|\mathbf{x}_{ji}, \widehat{u}_{ji})$  is  $S(t|\mathbf{x}_{ji}\widehat{\beta} + \widehat{u}_{ji})$ , where  $\widehat{u}_{ji}$  are the empirical Bayes predictions for  $u_{ji}$ . If `conditional(fixedonly)` is specified, then 0 is substituted for  $\widehat{u}_{ji}$ .

mean:

$$newvar_{ji} = \int_0^{\infty} \widehat{S}(t|\mathbf{x}_{ji}, u_{ji}) dt$$

surv:

$$newvar_{ji} = \widehat{S}(t_{ji}|\mathbf{x}_{ji}, \widehat{u}_{ji})$$

hazard:

$$newvar_{ji} = \widehat{g}(t_{ji}|\mathbf{x}_{ji}, \widehat{u}_{ji}) / \widehat{S}(t_{ji}|\mathbf{x}_{ji}, \widehat{u}_{ji})$$

where  $\widehat{g}(t|\mathbf{x}_{ji}, u_{ji})$  is the density  $g(t|\mathbf{x}_{ji}\widehat{\beta} + \widehat{u}_{ji})$ .

When the `marginal` option is used with `mean` or `surv`, the prediction is computed marginally with respect to the random effects. That is, the prediction is integrated over the random-effects distributions. When the `marginal` option is used with `hazard`, the hazard for the marginal distribution is computed. That is, the predicted hazard is computed as the quotient of the marginal hazard and the marginal survivor function.

## Also see

[ME] **mestreg** — Multilevel mixed-effects parametric survival models

[ME] **meglm postestimation** — Postestimation tools for meglm

[ME] **mixed postestimation** — Postestimation tools for mixed

[ST] **stcurve** — Plot survivor, hazard, cumulative hazard, or cumulative incidence function

[U] **20 Estimation and postestimation commands**